
1. PowerPC Assembly

• OS X and Linux different ABI

→ Linux less bone-headed

• OS X and Linux slightly differ-

ent assembly:

→ OS X less bone-headed

– OS X calls iregs r0-r31 and

fpregs f0-f31 and

– Linux calls iregs 0-31 and

fpregs 0-31

! Constants 0-31 are also 0-

31

⇒ Can use cpp tricks to work

around naming differences

• Suffixing many ops with ‘.’

makes them update CR

• PPC has big-endian byte and bit

ordering

– Sign bit bit 0 in first byte

– most sig bit (MSB) at bit 0

– LSB at bit 63

– low32 bits = 32-64

– hi32 bits = 0-31

– bit 0 at (1¡¡63)

• I’m going to use little endian bit

ordering, but PPC docs and ex-

plicit bit numbers assume big



2. CPP trick for Mixed OS X/Linux Assembly

• Need to rename registers for

Linux as shown at right

• Must have differing pro-

logue/epilogue/reg usage to

respect differing ABI

#ifdef ATL_GAS_LINUX_PPC

#define r0 0

#define r1 1

...

#define r31 31

#define f0 0

#define f1 1

...

#define f31 31

#endif



3. OS X PPC Register Usage

CALLEE
REGISTER USAGE SAVE

Integer Registers

r0∗ Used in prolog/epilog NO
r1 Stack pointer YES
r2 TOC pointer (reserved) YES
r3 1st para/return NO
r4 2nd para/return NO
r5-r10 3-8th para NO
r11 Environment pointer NO
r12 Used by global linkage NO
r13-31 Global int registers YES

Floating Point Registers

f0 Scratch reg NO
f1-13 1-13th fp para NO
f14-f31 Global fp regs YES

LR,CTR,XER special regs NO
CR[0-1,5-7] condition regs NO
CR[5-7] condition regs YES

• r0 interp as 0 in

EA calculations

• If fp para, the

appropriate

number of

para-passing

iregs are

skipped



4. OS X Calling Sequence and Stack Frame

• Stack grows downward

• Caller puts callees’ args in its

frame

• Frame 16-byte aligned

• Caller saves LR and CR only if

it wants to

• 8 word mandatory para area

not init (in reg); there for

callee use

• Has 224-byte red zone

Stack frame passed to callee

fp reg save area (optional)
ireg save area (optional)
padding (optional)
Local storage (optional)

24(r1) Parameter area (>= 8 words)
20(r1) TOC save area
16(r1) Link editor doubleword
12(r1) Compiler doubleword
8(r1) Link register (LR) save
4(r1) Condition register (CR) save
0(r1) ptr to callee’s stack



5. Linux PPC Register Usage

CALLEE
REGISTER USAGE SAVE

Integer Registers

r0∗ Used in prolog/epilog NO
r1 Stack pointer YES
r2 TOC pointer (reserved) YES
r3-r4 1/2 para and return NO
r5-r10 3-8th integer para NO
r11-r12 Func linkage regs NO
r12 Used by global linkage NO
r13 Small data area ptr reg NO
r14-30 Global int registers YES
r31 Global/environment ptr YES

Floating Point Registers

f0 Scratch reg NO
f1 1st para / return NO
f2-8 2-8th fp para NO
f9-f13 Scratch reg NO
f14-f31 Global fp regs YES

• r0 interp as 0 in

EA calculations

• fp para, does not

cause skip of iregs



6. Linux PPC Calling Sequence and Stack Frame

• Stack pointer 16-byte aligned.

• Stack pointer (sp) updated atomically
via “store word with update”.

• Any non-scratch reg f# must be saved
to the fp reg save area at location 8*(32-
#) from the previous frame (i.e., the
register f31 is saved adjacent to previous
ptr to callee’s stack).

• Any non-scratch reg r# must be saved
in the ireg save area 4*(32-#) bytes be-
fore the low-addressed end of the fp reg
save area.

• Minimum stack frame callee save and
LR save.

• No red zone is specifically mandated.

• If number if ipara is ≤ 8 and the num-
ber of fppara ≤ 8, then no values are
stored in the parameter area, and if this
is true for all calls made by the routine,
the parameter area will be of size 0.

Stack frame passed to callee

fp reg save area (optional)
ireg save area (optional)
CR save area (optional)
Local storage (optional)

8(r1) Parameter area (optional)
4(r1) Link register (LR) save
0(r1) ptr to callee’s stack



7. PowerPC System Registers

1. CR: Condition Register - 32

bits, set implicitly or by [f]cmp

• mfcr rx: rx0−31 = CR;

rx32−63 = 0

• mtcr rx: CR = rx0−31

• Have bunch of boolean inst

as well

2. CTR: Count Register

• mfctr rx: rx0−31 = CTR;

rx32−63 = 0

• mtctr rx: CTR = rx0−31

3. LR: Link Register - Used to save

address to return to

• mflr rx: rx = LR

• mtlr rx: LR = rx

4. XER: Fixed-pt Exception Reg-

ister

• mfxer rx: rx0−31 = XER;

rx32−63 = 0

• mtxer rx: XER = rx0−31

5. FPSCR: FP Status & Control

Register

• mffs rx: rx0−31 = FPSCR;

rx32−63 = undef

• mtfs rx: FPSCR = rx0−31

• Other bit/mask forms as well

6. MSR: Machine State Register

• mfmsr rx: rx = MSR

• mtmsr rx: MSR = rx



8. PowerPC Assembly Overview

• Three-operand assembly (some 4-op mnem):

– <mnem> <dest>, <src1>, <src2>

• ld/st only inst that take addresses

• inst that take immediates end in i (eg, andi).

– Most inst take signed 16-bit immediates

• Both 16-bit imm & 32-bit reg are sign extended for most ops

• Other inst have only reg operands

• No prefix for reg or constants!

• To make inst affect CR, suffix them with ‘.’

• Inst that are cracked (dec into 2 internal inst) on G5 marked byc

– take up extra room in group

• Inst that are microcoded (decoded into > 2 inst) marked bym

– Only 1 microcoded inst can be fetched from inst fetch buff into

internal pipe per cycle



9. PowerPC Addressing Modes

Two general forms of addressing:

1. Offset: imm16(rx)

• @ = rx + imm16

2. Indexed : rx, ry

• @ = rx + ry

!! r0 interp as 0 (except when ry):

• 64(r0): @ = 64

• r0, r1: @ = r1

• r1, r0: @ = r1+r0

Two types of ld to handle 64 bits

1. Load and zero

• load to lower 32 bits, clear

upper

• suffix : z

2. Load Algebraic

• load to lower 32 bits, sign ex-

tend upper

• suffix : a



10. PPC Integer Transfer Operations

Mnemonic Operands Action

lwz rz, i(rx) rz0−31 = *(rx+i), rz32−63 = 0
lwzuc rz, i(rx) rz0−31 = *(rx+i), rz32−63=0, rx += i
lwzx rz, rx, ry rz0−31 = *(rx+ry), rz32−63 = 0
lwzuxm rz, rx, ry rz0−31 = *(rx+ry), rz32−63 = 0, rx += ry

lwac rz, i(rx) rz0−31 = *(rx+i), rz32−63 = *(rx+i)31
lwaum rz, i(rx) rz0−31 = *(rx+i), rz32−63=*(rx+i)31, rx += i
lwaxc rz, rx, ry rz0−31 = *(rx+ry), rz32−63 = *(rx+ry)31
lwauxm rz, rx, ry rz0−31 = *(rx+ry), rz32−63 = *(rx+ry)31, rx += ry

stw rz, i(rx) *(rx+i) = rz0−31
stwuc rz, i(rx) *(rx+i) = rz0−31, rz += i
stwxc rz, rx, ry *(rx+ry) = rz0−31
stwuxm rz, rx, ry *(rx+ry) = rz0−31, rx += ry

mr rz, rx rz = rx (register copy)
li rz, i rz0−15 = i, rz16−63 = 0
lis rz, i rz0−15 = 0, rz16−31 = i, rz32−63 = 0

• Setting rx as r0 substitutes zero for rx. (illegal with update forms)

• Yes, STs take last arg as dest!



11. PPC 64 bit Integer LD/ST Operations

Mnemonic Operands Action

ld rz, i(rx) rz = *(rx+i),
lduc rz, i(rx) rz = *(rx+i), rx += i
ldx rz, rx, ry rz = *(rx+ry)
lduxm rz, rx, ry rz = *(rx+ry), rz = 0, rx += ry

std rz, i(rx) *(rx+i) = rz
stduc rz, i(rx) *(rx+i) = rz, rz += i
stdx rz, rx, ry *(rx+ry) = rz
stduxm rz, rx, ry *(rx+ry) = rz, rx += ry

• Setting rx as r0 substitutes zero for rx. (illegal with update forms)



12. Common Integer Arithmetic Operations

Mnemonic Operands Action

add[o][.] rz, rv, ry rz = rv + ry
addcc[om][.m] rz, rv, ry rz = rv + ry, set XER[CA] if carry past 32 bits

addi rz, rx, i rz = rx + i
addis rz, rx, i rz = rx + (i<<16)
addic[.c] rz, rx, i rz = rx + i set XER[CA] if carry past 32 bits

sub[oc][.] rz, rv, ry rz = rv - ry
subf[oc][.] rz, rv, ry rz = ry - rv
subfcc[om][.m] rz, rv, ry rz = ry + rv, set XER[CA] if carry past 32 bits
subfic[o][.] rz, rv, i rz = i - rv, set XER[CA] if carry past 32 bits

neg[o][.] rz, ry rz = -ry
extsw[.c] rz, ry rz0−31 = ry0−31, rz32−63 = ry31
cntlzw[.] rz, ry rz = leading zeros(ry0−31)
cntlzd[.] rz, ry rz = leading zeros(ry)

Suffixes:

• ’.’: set CR[0] to indicate >, <, =

• ’o’: inst sets overflow (XER[OV])

• if rx is r0, means zero

• i sign extended to 64 bits

• nego. cracked



13. Common Integer Multiply/Divide Operations

Mnemonic Operands Action

mullw[o][.c] rz, rv, ry rz = rv0−31 * ry0−31
mulld[o][.c] rz, rv, ry rz = (rv * ry)0−63
mulhd[.c] rz, rv, ry rz = (rv * ry)64−127
mulhdu[.c] rz, rv, i rz = (rv * ry)64−127 (unsigned)
mulli rz, rv, i rz = rv * i

divwc[om][.m] rz, rv, ry rz = rv0−31 / ry0−31
divwuc[om][.m] rz, rv, ry rz = rv0−31 / ry0−31 (unsigned)
divdc[om][.m] rz, rv, ry rz = rv / ry
divduc[om][.m] rz, rv, ry rz = rv / ry (unsigned)

Suffixes:

• ’.’: set CR[0] to indicate >, <, =

• ’o’: inst sets overflow (XER[OV])



14. Common PPC Boolean Bit-Level Operations

Mnemonic Operands Action

and[.] rz, rv, ry rz = rv & rz
andc[.] rz, rv, ry rz = rv & (~rz)
eqv[.] rz, rv, ry bit in rz set if same bit in rx and ry match
nand[.] rz, rv, ry rz = ~(rv & rz)
nor[.] rz, rv, ry rz = ~(rv | rz)
or[.] rz, rv, ry rz = rv | rz
orc[.] rz, rv, ry rz = rv | (~rz)
or[.] rz, rv, ry rz = rv ^ rz

andi. rz, rv, i rz0−15 = rv & i, rz16−63 = 0
andis. rz, rv, i rz16−31 = rv & (i<<16), rz32−63 = rz0−15 = 0
ori rz, rv, i rz = rv | i
oris rz, rv, i rz = rv | (i<<16)
xori rz, rv, i rz = rv ^ i
xoris rz, rv, i rz = rv ^ (i<<16)

• Only ops ending in ’.’ update CR[0]



15. PPC Shift Operations

Mnemonic Operands Action

slw[.c] rz, rv, ry rz0−31 = (rv0−31 << ry); rz32−63 = 0
sld[.c] rz, rv, ry rz = (rv << ry)
slwi[.] rz, rv, i rz0−31 = (rv0−31 << i); rz32−63 = 0
sldi[.] rz, rv, i rz = (rv << i)

srw[.c] rz, rv, ry rz0−31 = (rv0−31 >> ry); rz32−63 = 0
srd[.c] rz, rv, ry rz = (rv >> ry)
srwi[.] rz, rv, i rz0−31 = (rv0−31 >> i); rz32−63 = 0
srdi[.] rz, rv, i rz = (rv >> i)

sraw[.c] rz, rv, ry rz0−31 = (rv0−31 >> ry); rz32−63 = 0, dup sign bit
srad[.c] rz, rv, ry rz = (rv >> ry), dup sign bit
srawi[.c] rz, rv, i rz0−31 = (rv0−31 >> i); rz32−63 = 0, dup sign bit
sradi[.c] rz, rv, i rz = (rv >> i), dup sign bit

• Also have mask & rotate inst

• Also have Extract & justify inst

⇒ Very useful, but we’re not covering them.



16. PPC Condition Codes

Conditions signaled in condition

register (CR):

• 8 different 4-bit CRs in 32-bit

CR:

– CR[0] implicit CR for iops

– CR[1] implicit CR for fops

– CR[2-7] used explicitly

• Int CR bits (big endian):

0. LT: set if rv < ry

1. GT: set if rv > ry

2. EQ: set if rv = ry

3. SO: set if overflow

• FP CR bits (big endian):

0. FL: set if rv < ry

1. FG: set if rv > ry

2. FE: set if rv = ry

3. FU: set if unordered



17. Common PPC Comparison Instructions

Mnemonic Operands Action

cmpw crz, rv, ry set CR[z] rv0−31 ? ry0−31
cmplw crz, rv, ry set CR[z] rv0−31 ? ry0−31 (unsigned)
cmpwi crz, rv, i set CR[z] rv0−31 ? sign ext(i)
cmplwi crz, rv, i set CR[z] rv0−31 ? zero ext(i) (unsigned)

cmpd crz, rv, ry set CR[z] rv ? ry
cmpld crz, rv, ry set CR[z] rv ? ry (unsigned)
cmpdi crz, rv, i set CR[z] rv ? sign ext(i)
cmpldi crz, rv, i set CR[z] rv ? zero ext(i) (unsigned)

• Can do comparison(s) early, branch later

→ Can store multiple results in differing CR

– Implicit icmp set CR[0]

– Implicit fcmp set CR[1]



18. Common PPC Branch Instruction

Mnemonic Operands Action

blr[l] none (ret/call) jump to @ in LR

b[l] label unconditional jump

beq[l][a] [crx,] label jump if CR[x][EQ] = 1
bne[l][a] [crx,] label jump if CR[x][EQ] = 0
blt[l][a] [crx,] label jump if CR[x][LT] = 1
ble[l][a] [crx,] label jump if CR[x][GT] = 0
bgt[l][a] [crx,] label jump if CR[x][GT] = 1
bge[l][a] [crx,] label jump if CR[x][LT] = 0
bso[l][a] [crx,] label jump if CR[x][SO] = 1
bun[l][a] [crx,] label jump if CR[x][FU] = 1
bnu[l][a] [crx,] label jump if CR[x][FU] = 0
bdnz[l][a] label jump if (−−CTR) 6= 0

• bne+ 3, LOOP

– Jump to LOOP label based on CR[3], jump

likely taken

• bgtla- r2

– Jump @ in r2 based on CR[0], jump unlikely

taken, save @ of next inst in LR

• ‘l’: store next inst

@ in LR

• ‘a’: label is abso-

lute address (of-

ten in register)

• Can also give

branch prediction

suffix hint:

– ’+’: branch

likely to be

taken

– ’-’: branch

unlikely to be

taken

• Lot more br

types!



19. PPC Floating Point

• FPU is all double precision, single load/store converts

• Computational fpinst take ’s’ suffix, to indicate single precision:

→ Causes single rounding for double register

– Could instead use double computation, and convert only at end

for extra precision

• Most instructions can be suffixed with ‘.’ (not too useful most of

the time IMHO) so that CR[1] indicates exception status, as shown

below.

CR[1] Name Description

0 FX Set on any exception
1 FEX Set on any enabled exception
2 VX Set on any invalid operation exception
3 OX Set on overflow



20. PPC Floating Point Load/Stores/Move

Mnemonic Operands Action

lf[d,s] frz, i(rx) frz = *(rx+i)
lf[d,s]uc frz, i(rx) frz = *(rx+i), rx += i
lf[d,s]x frz, rx, ry frz = *(rx+ry)
lf[d,s]uxc frz, rx, ry rz = *(rx+ry), rx += ry

stf[d,s] frz, i(rx) *(rx+i) = frz
stf[d,s]uc frz, i(rx) *(rx+i) = frz, rx += i
stf[d,s]x frz, rx, ry *(rx+ry) = frz
stf[d,s]uxc frz, rx, ry *(rx+ry) = frz, rx += ry

fmr[.] frz, frx frz = frx (register copy)
fabs[.] frz, frx frz = |frx|
fnabs[.] frz, frx frz = −|frx|
fneg[.] frz, frx frz = −frx
frsp[.] frz, frx frz = Double2Float(frx)

• ‘s’: convert float

to double on ld/st

• ‘d : just load dou-

ble fpnum into reg



21. Common PPC Floating Point Computation Instructions

Mnemonic Operands Action

fmadd[s][.] frz, frv, frx, fry frz = frv * frx + fry
fmsub[s][.] frz, frv, frx, fry frz = frv * frx - fry
fnmadd[s][.] frz, frv, frx, fry frz = -(frv * frx + fry)
fnmsub[s][.] frz, frv, frx, fry frz = -(frv * frx - fry)

fadd[s][.] frz, frx, fry frz = frx + fry
fsub[s][.] frz, frx, fry frz = frx - fry
fmul[s][.] frz, frx, fry frz = frx * fry
fdiv[s][.] frz, frx, fry frz = frx / fry

fcmpo crz, frx, fry cmp, any NaN a NaN
fcmpu crz, frx, fry cmp, only signalling NaN a NaN



22. Simple DAXPY in PPC Assembly

r3 f1:r4/r5 r6 r7 r8 r9
void ATL_UAXPY(int N, double alpha, double *X, int incX, double *Y, int incY)

Moving Ptrs:
#define Mjoin(pre, nam) my_join(pre, nam)
#define my_join(pre, nam) pre ## nam
#define N r3
#define X r6
#define Y r8
#define alpha f1
#define rX f2
#define rY f3

.text

.globl Mjoin(_,ATL_UAXPY)
Mjoin(_,ATL_UAXPY):
LOOP:

lfd rX, 0(X)
lfd rY, 0(Y)
fmadd rY, alpha, rX, rY
stfd rY, 0(Y)
addi X, X, 8
addi Y, Y, 8
addic. N, N, -1
bne+ LOOP

DONE:
blr

Indexed:
Mjoin(_,ATL_UAXPY):

sldi II, N, 3
add X, X, II
add Y, Y, II
neg II, II

LOOP:
lfdx rX, X, II
lfdx rY, Y, II
fmadd rY, alpha, rX, rY
stfdx rY, Y, II
addic. II, II, 8
bne+ LOOP

DONE:
blr



23. Simple SASUM

/* r3 r4 r5
float ATL_UASUM(int N, float *X, const int incX) */

#define Mjoin(pre, nam) my_join(pre, nam)
#define my_join(pre, nam) pre ## nam
#define II r3
#define X r4
#define rsum f1
#define rX f2

.text

.globl Mjoin(_,ATL_UASUM)
Mjoin(_,ATL_UASUM):

xor r5, r5, r5
stw r5, -4(r1)
lfs rsum, -4(r1)
sldi II, II, 2
add X, X, II
neg II, II

LOOP:
lfsx rX, X, II
fabs rX, rX
fadds rsum, rsum, rX
addic. II, II, 4
bne+ LOOP

DONE:
blr

#define Mjoin(pre, nam) my_join(pre, nam)
#define my_join(pre, nam) pre ## nam
#define N r3
#define X r4
#define rsum f1
#define rX f2

.text

.globl Mjoin(_,ATL_UASUM)
Mjoin(_,ATL_UASUM):

xor r5, r5, r5
stw r5, -4(r1)
lfs rsum, -4(r1)
mtctr N

LOOP:
lfs rX, 0(X)
fabs rX, rX
fadds rsum, rsum, rX
addi X, X, 4
bdnz+ LOOP

DONE:
blr


